



**ESKİŞEHİR OSMANGAZİ ÜNİVERSİTESİ**

Mühendislik Mimarlık Fakültesi

İnşaat Mühendisliği Bölümü

E-Posta: [ogu.ahmet.topcu@gmail.com](mailto:ogu.ahmet.topcu@gmail.com)

Web: <http://mmf2.ogu.edu.tr/atopcu>

**Bilgisayar Destekli**

# **Nümerik Analiz**

*Ders notları 2014*

**Ahmet TOPÇU**

```
C:\Basic\QBASIC.EXE
File Edit View Search Run Debug Calls Utility Options Help
GAJORDSV.BAS:GaussJorSim
DEFINT I-N
DEFDBL A-H, O-Z
SUB GaussJorSim_(n, a(), iHata)
'-----
' Ters matris hesabı
' Dr. Ahmet TOPÇU, Osmangazi Üniversitesi, ESKİŞEHİR, 1995
' a(n,n) simetrik ve pozitif tanımlı matrisin tersi
' hesaplanarak a(n,n) üzerinde depolanır.
' Pivot arama yapılmaz.
' a çağırılan programda depolanmış olmalıdır.
' iHata=0 ise ters matris hesaplanmıştır.
' iHata>0 ise matris tekildir, tersi hesaplanmaz.
'-----
' Machep
Eps = 1
DO
  Eps = Eps / 2
  s = 1 + Eps
LOOP UNTIL s <= 1
Eps = 2 * Eps

  iHata = 0
  Zero = 0
  FOR i = 1 TO n
    IF ABS(a(i, i)) >= Zero THEN Zero = ABS(a(i, i))
  NEXT i
  Zero = Zero * Eps

  FOR i = 1 TO n
    IF ABS(a(i, i)) <= Zero THEN iHata = 1: EXIT SUB
    T = -1 / a(i, i)
    a(i, i) = -1
    FOR k = 1 TO n
      a(k, i) = a(k, i) * T
    NEXT k
  ' Çarpım
  FOR k = 1 TO n
    IF k <> i THEN
      T = a(i, k)
      a(i, k) = 0
      FOR j = 1 TO n
        a(j, k) = a(j, k) + a(j, i) * T
      NEXT j
    END IF
  NEXT k
END SUB
```

# 19

**PROGRAMLAR:**

Simetrik Ters matris (GAUSS-JORDAN tekniği)

## 19. PROGRAMLAR: Simetrik ters matris hesabı-GaussJorSim

Sayısal hesaplarda ters matrisin teorik olarak önemi büyük olmakla birlikte, zorunlu olmadıkça hesabından kaçınılır. Hesaplanması zorunlu durumlarda aşağıda kodu verilen GaussJorSim alt programı simetrik matrisler için kullanılabilir. Simetrik ters matris COLESKY metodu ile de hesaplanabilir, ancak GAUSS-JORDAN metodundan iki kat daha fazla ana bellek gerektirir. Ters matris daima tam dolu bir matris olduğundan bant özelliğinden yararlanılamaz.

GaussJordan alt programı

$$\underline{A}_{n \times n} = \begin{bmatrix} a_{11} & a_{12} & \dots & a_{1n} \\ & a_{22} & \dots & a_{2n} \\ \cdot & \cdot & \dots & \cdot \\ \text{Simetrik} & & \dots & a_{nn} \end{bmatrix}$$

matrisinin  $\underline{A}_{n \times n}^{-1}$  ters matrisini GAUSS-JORDAN tekniği ile hesaplar. Pivot eleman satırda aranır ve gerekirse kolonlara yer değiştirilir. Ters matris  $\underline{A}$  nın üzerine depolanır. Program iHata $\neq$ 0 ile dönerse det  $\underline{A}=0$  dır, ters matris hesaplanmaz.

**Örnek:**

$$\underline{A} = \begin{bmatrix} 4 & 2 & 4 & 2 & 4 \\ 2 & 5 & 4 & 4 & 4 \\ 4 & 4 & 6 & 2 & 2 \\ 2 & 4 & 2 & 9.5 & 2 \\ 4 & 4 & 2 & 2 & 32 \end{bmatrix}, \quad \underline{A}^{-1} = ?$$

GaussJorSim sonucu

```

C:\Basic\QBASIC.EXE
Ters matris(GaussJorSim):
3.90625 3.020833333333333 -4.0625 -1.125 -.541666666666667
3.020833333333333 3.069444444444444 -3.541666666666667 -1.083333333333333
-4.722222222222222
-4.0625 -3.541666666666667 4.625 1.25 .583333333333333
-1.125 -1.083333333333333 1.25 .5 .166666666666667
-.541666666666667 -.472222222222222 .583333333333333 .166666666666667
.111111111111111
  
```

**Çözüm:**

$$\underline{A}^{-1} = \begin{bmatrix} 3.9063 & 3.0208 & -4.0625 & -1.2250 & -0.5417 \\ 3.0208 & 3.0694 & -3.5417 & -1.0833 & -0.4722 \\ -4.0625 & -3.5417 & 4.6250 & 1.2500 & 0.5833 \\ -1.1250 & -1.0883 & 1.2500 & 0.5000 & 0.1667 \\ -0.5417 & -0.4722 & 0.5833 & 0.1667 & 0.11111 \end{bmatrix}$$

```

'----- Ana program GaussJorSim-----
' Simetrik matrisin tersinin hesabı
' Çağrılan alt programlar: GaussJorSim
'-----
DATA 5: ' Matrisin boyutu
' A matrisi
DATA 4, 2, 4, 2, ,4
DATA 5, 4, 4, ,4
DATA 6, 2, ,2
DATA 9.5, 2
DATA 32

DEFINT I-N
DEFDBL A-H, O-Z

DECLARE SUB GaussJorSim (n, a(), iHata)

CLS
READ n
DIM a(n, n)
' matrisinin üst üçgenini oku, simetriye tamamla
FOR i = 1 TO n
FOR j = i TO n
READ a(i, j)
a(j, i) = a(i, j)
NEXT j
NEXT i

CALL GaussJorSim(n, a(), iHata)

IF iHata <> 0 THEN PRINT "Matris tekil (GaussJorSim)": END

PRINT "Ters matris(GaussJorSim):"
FOR i = 1 TO n
FOR j = 1 TO n
PRINT a(i, j);
NEXT j
PRINT
PRINT
NEXT i

END 'GaussJorSim ana

SUB GaussJorSim (n, a(), iHata)
'-----
' Ters matris hesabı
' Dr. Ahmet TOPÇU, Osmangazi Üniversitesi, ESKİŞEHİR, 1995
' a(n,n) simetrik ve pozitif tanımlı matrisin tersi
' hesaplanarak a(n,n) üzerinde depolanır.
' Pivot arama yapılmaz.
' a çağırılan programda depolanmış olmalıdır.
' iHata=0 ise ters matris hesaplanmıştır.
' iHata<>0 ise matris tekildir, tersi hesaplanmaz.
'-----
' Machepe
Eps = 1
DO
Eps = Eps / 2
s = 1 + Eps
LOOP UNTIL s <= 1
Eps = 2 * Eps

iHata = 0
Zero = 0
FOR i = 1 TO n
IF ABS(a(i, i)) >= Zero THEN Zero = ABS(a(i, i))
NEXT i
Zero = Zero * Eps
FOR i = 1 TO n
IF ABS(a(i, i)) <= Zero THEN iHata = 1: EXIT SUB
T = -1 / a(i, i)
a(i, i) = -1
FOR k = 1 TO n
a(k, i) = a(k, i) * T
NEXT k
' Çarpım
FOR k = 1 TO n
IF k <> i THEN
T = a(i, k)
a(i, k) = 0
FOR j = 1 TO n
a(j, k) = a(j, k) + a(j, i) * T
NEXT j
END IF
NEXT k
NEXT i
END SUB 'GaussJorSim

```

A'nın satırları

Ana program

A'nın boyutu okunuyor

A matrisi için bellekte yer ayır

A okunuyor

GAUSSJORSim alt programını çağır

Sonucu yaz

GAUSSJORSim  
Alt programı